Getting Started with the MSP430 IAR Assembly

by Alex Milenkovich, milenkovic@computer.org

Objectives: This tutorial will help you get started with the MSP30 IAR Assembly program development. You will learn the following topics:

- Assembly programming
- Creating an application project using assembly programs
- Debugging using the IAR C-SPY® Debugger (Simulator)

Note: It is required that students have completed the tutorial *Getting Starting With MSP430 IAR Embedded Workbench* before starting with this one.

1. **Count Number of Characters ‘E’ in a String: Problem Statement**

This section defines the problem that will be solved by the *Count Characters* program using MSP430 assembly language.

Our task is to develop an assembly program that will scan a given string of characters, for example, “HELLO WORLD, I AM THE MSP430!”, and find the number of appearances of the character ‘E’ in the string. A counter that records the number of characters ‘E’ is then written to the parallel port P1. The port should be configured as an output port, and the binary value of the port will correspond to the counter value.

Step 1: Analyze the assignment and develop a plan.

To solve this assignment, let us first analyze the problem statement. Your task is to write an assembly program that will count the number of characters ‘E’ in a string. First, the problem implies that we need to allocate space in memory that will keep the string “HELLO WORLD, I AM THE MSP430!”.

The string has 29 characters and they are encoded using ASCII table. To allocate and initialize a string in memory we can use an assembly language directive, for example DB (Define Byte).

```
DB "HELLO WORLD, I AM THE MSP430!"
```

We can also put a label to mark the beginning of this string in memory, for example, mystr:

```
mystr DB "HELLO WORLD, I AM THE MSP430!"
```

When assembler sees the DB directive, it will allocate the space in memory required for the string that follows and initialize the allocated space with ASCII characters. The assembler will also automatically add an ASCII NULL character at the end of each string (ascii(NULL)=0x00). So, the total number of bytes occupied by this string is 30.
Step 2. Develop a plan.
Our task is now to write a small program that will scan the string, character by character, check whether the current character is equal to character ‘E’, and if yes, increment a counter. The string scan is done in a program loop. The program ends when we reach the end of the string which is detected when the current character is equal to the NULL character (0x00).
To scan the string we will use a register to point to the current character in the string. This pointer register is initialized at the beginning of the program to point to the first character in the string. The pointer will be incremented in each iteration of the program loop. Another register, initialized to zero at the beginning, will serve as the counter, and it is incremented every time the current character is ‘E’.
After we exit the program loop, the current value of the counter will be written to the port P1, which should be initialized as an output port.

Note: it is required that you are familiar with the MSP430 instruction set and addressing modes to be able to solve this problem. We assume that the string is no longer than 255 characters, so the result can be displayed on an 8-bit port.

2. Count Characters: Assembly Code
Figure 1 shows the assembly code for this program. Here is a short description of the assembly code.
1. The comments in a single line start with a column character (;). Multi-line comments can use C-style /* comment */ notation.
2. #include <msp430xG46x.h>; This is a C-style pre-processor directive that specifies a header file to be included in the source. The header file includes all macro definitions, for example, special function register addresses (WDTCTL), and control bits (WDTPW+WDTHOLD).
3. Use ORG assembly directive to set the program location counter of the current segment to the value of an expression that follows. Here ORG 0FF00h sets the location counter at the absolute address 0FF00h. This means that location counter is moved to this address.
4. Next, we allocate the string myStr that will start at the location 0FF00h using DB directive: myStr DB "HELLO WORLD, I AM THE MSP430!". As explained, this directive will allocate 30 bytes in memory starting at the address 0FF00h and initialize it with the string content. The content in memory will be as follows: 48 45 4c 4c 4f 20 57 4f 52 4c 44 2c 20 49 20 41 4d 20 48 45 20 4d 53 50 34 33 30 21 00.
5. How does my program execute on an MSP430? Upon powering-up the MSP430 control logic always generates a RESET interrupt request (it is the highest priority interrupt request). The value stored at the address 0xFFFE (the last word in the 64KB address space) is reserved to keep the starting address of the reset handler (interrupt service routine), and the first thing that the microcontroller does is to fetch the content from this address and put it in the program counter (PC, register R0). Thus, the starting address of our program should be stored at location 0xFFFE. Here, we move location counter to 0xFFFE and allocate 2 bytes (DC16 allocates 16 bits or two bytes) that will be initialized with the starting address of the main program. The starting address of the main program is marked by the label init.
6. RSEG is a segment control assembler directive that controls how code and data are located in memory. RSEG is used to mark the beginning of a relocatable code or data segment. CODE
and DATA are recognized segment types that are resolved by the linker. The IAR XLINK linker can recognize any other type of segment (e.g., CSTACK for code stack).

7. First instruction initializes the stack pointer register (MOV #SFE(CSTACK), SP). Our program does not use the stack, so we could have omitted RSEG CSTACK and this instruction.

8. The instruction MOV.W #WDTPW+WDTHOLD, &WDTCTL sets certain control bits of the watchdog timer control register (WDTCTL) to disable it. The watchdog timer by default is active upon reset, generating interrupt requests periodically. As this functionality is not needed in our program, we simply need to disable it.

9. Parallel ports in MSP430 microcontroller can be configured as input or output. A control register PxDIR determines whether the port x is input or output (we can configure each individual port pin). Our program drives all pins of the port P1, so it should be configured as the output (P1DIR=0xFF). Register R4 is loaded to point to the first character in the string. Register R5, the counter, is cleared before starting the main program loop.

10. The main loop starts at the gnext label. We use autoincrement addressing mode to read a new character (one byte) from the string (MOV.B @R4+, R6). The current character is kept in register R6. We compare the current character with the NULL character (CMP #0, R6). If it is the NULL character it is the end of the string and we exit the loop (JEQ lend). Pay attention that we used JEQ instruction? Why?
    If it is not the end of the string, we compare the current character with ‘E’. If there is no match we go back on the first instruction in the loop. Otherwise, we increase the value of the counter (register R5).

11. Finally, we move the lower byte from R5 to the parallel port 1, P1OUT=R5[7:0].
/*---------------------------------------------
* Program    : Counts the number of characters E in a string
* Input      : The input string is the myStr
* Output     : The port one displays the number of E's in the string
* Written by : A. Milenkovic
* Date       : August 14, 2008
* Description: MSP430 IAR EW; Demonstration of the MSP430 assembler
*---------------------------------------------*

#include "msp430.h"          ; #define controlled include file

ORG  OFF00h

myStr   DB "HELLO WORLD, I AM THE MSP430!" ; the string is placed on the stack
; the null character is automatically added after the '!

NAME    main                    ; module name
PUBLIC  main                    ; make the main label visible
; outside this module
ORG     OFFFEh

DC16    init                    ; set reset vector to 'init' label
RSEG    CSTACK                  ; pre-declaration of segment
RSEG    CODE                    ; place program in 'CODE' segment

init:   MOV     #SFE(CSTACK), SP ; set up stack
main:   NOP                             ; main program
MOV.W   #WDTPW+WDTHOLD,&WDTCTL  ; Stop watchdog timer
BIS.B   #0FFh,&P1DIR            ; configure P1.x output
MOV.W   #myStr, R4                ; load the starting address of the string into the
register R4
CLR.B   R5                        ; register R5 will serve as a counter
gnext:  MOV.B   @R4+, R6                  ; get a new character
CMP     #0,R6
JEQ     lend                        ; go to the end
CMP.B   #'E',R6
JNE     gnext
INC     R5                          ; increment counter
JMP     gnext

lend:   MOV.B   R5,&P1OUT                ; Set all P1 pins
BIS.W   #LPM4,SR                  ; LPM4
NOP                  ; Required only for debugger

END

Figure 1. MSP430 Assembly Code for Count Character Program.
3. Creating an Application Project

Please consult the tutorial on how to create a new workspace and project.

Step 1. Choose Project>Create New Project. Select Empty project and save it as char_count.

Step 2. Add the source file char_count.s43 to the project. Choose Project>Add Files.

Step 3. Choose Project>Options and set all categories as explained in the Getting Starting With MSP430 IAR Embedded Workbench tutorial.

Step 4. Select the source file char_count.s43 and compile it. (Choose Project>Compile). Click on the list file char_count.lst to see report generated by the compiler.

Step 5. Link the program. Choose Project>Make. The application is ready for debugging.

4. Program Simulation and Debugging

In this section we will discuss how to simulate program execution.

Step 1: Go to Project>Options and select Debugger category. In the Driver box, choose Simulator (Figure 2). Click OK.

Step 2: Choose Project>Debug. You will see the EW display as shown in Figure 3.

Step 3: Step through the program step by step (use F11). Observe the Disassembly, Register View, and Memory View windows. Answer the following questions.
What is the starting address of the program?
How many clock cycles does each instruction take to execute?
Observe the contents of memory location and registers as you step through the program. What is the content of the memory location at the address 0xFFFE?

What are addresses of the special-purpose registers P1DIR and P1OUT? Monitor the contents of these locations as you walk through your program. Set breakpoints to move easier through your program.

Step 4: Choose Stop Debugging.
Figure 2. Debugger category.

Figure 3. IAR Simulator.
5. Assignments

1. Write an assembly program that will create a new output string by processing an input string. Each lower case letter character in the input string should be replaced by the corresponding upper case letter character. Assume the input string is “Welcome to cpe323. Yours msp430!”. Count the number of characters in the string and the number of changes. Set the port P1 to display the total number of character, and port P2 to display the number of changes.